
1. [bookmark: _Toc122434485][bookmark: _Toc117597714]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 8.1.6.1.3.6 which is part of the 5G NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Shaun Harry
	shaun.harry@keysight.com
0. [bookmark: _Toc117597715]Verification Test Summary 
Test Case: 	8.1.3.1.3.6
ATS Version:	iwd-TTCN3-B2020-09_D22wk37
System Simulator used:	Keysight S8704A Protocol Conformance Toolset
UE used:	MediaTek MT6895
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc117597716]Corrections required
1. [bookmark: _Toc30685522][bookmark: _Toc117597717]Change 1
	Function name
	f_TC_8_1_6_1_3_6_NR5GC_TestBody

	Reason for change
	1. [bookmark: OLE_LINK4]Step 12 “The SS is configured to ignore the preambles transmitted by the UE on PRACH” is not implemented

2.  System indication for  PRACH preamble need to be enabled before step 14 

3. According to 38.321 5.1.2:
If the selected RA_TYPE is set to 4-stepRA, the MAC entity shall:
1> else if the ra-PreambleIndex has been explicitly provided by PDCCH; and 
[bookmark: OLE_LINK6]1> if the ra-PreambleIndex is not 0b000000: 
2> set the PREAMBLE_INDEX to the signalled ra-PreambleIndex; 
2> select the SSB signalled by PDCCH.
Since the ra-PreambleIndex at Step13 is configured to 0b000000, the PREAMBLE_INDEX used in step 14 shall depend on the UE selection. So, the ra-PreambleIndex cannot be checked at step14
4. v_Repetitions is Zero-based, so when the 7th PRACH preamble is received it will be 6, not 7
5. In the RLF report checking at step 21, the expected value of reestablishmentCellId_r16 is omit, but the UE may include this since reestablishment happened earlier in the sequence.

	Summary of change
	1. Configure SS to ignore RACH preambles at step 12

2. enable system indications for RACH preamble before step13 and disable  after step 14

3. set the expected ra-PreambleIndex to ? at step 14

Note : An associated prose CR on 38.523-1 will be raised at RAN5#97

4. change the condition to v_Repetitions == 6

5. allow reestablishmentCellId-r16 to be included in RLF report


	TTCN module
	RRC_MDT_IntraNR_RLF_NR5GC

	MCC160 Comment
	



Before Change
	  function f_TC_8_1_6_1_3_6_NR5GC_TestBody() runs on NR5GC_PTC
  { // @sic R5-223274 R5-225390 sic@
   var NR_SRB_COMMON_IND v_ReceivedAsp;
   var PLMN_Identity v_PlmnID_NRCell1 := f_NR_CellInfo_GetPLMN (nr_Cell1);
   var CellIdentity v_CellIdentity_NRCell1 := f_NR_CellInfo_GetCellIdentity(nr_Cell1);
   var TrackingAreaCode v_TAC_NRCell1 := f_NR_CellInfo_GetTAC (nr_Cell1);
   var RNTI_Value v_RNTI_Value := f_NR_CellInfo_GetRNTI(nr_Cell1);
   var RLF_Report_r16 v_Nr_Rlf_Report_r16;
   
   var template (value) NR_PDCCH_Order_Type v_PDCCH_Order;
   var integer v_RA_PreambleIndex := 0;
   var integer v_Repetitions := 0;
   timer t_Watchdog := 5.0;              // Local wait Timer

   //@siclog "Steps 1-11 - Void " siclog@
   
   //@siclog "Step 12" siclog@
   // The SS is configured to ignore the preambles transmitted by the UE on PRACH
   f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(disable));

   //@siclog "Step 13" siclog@
   // The SS sends a PDCCH order with ra-PreambleIndex = '000000'B.
   v_PDCCH_Order := cs_NR_PDCCH_Order(f_NR_CellInfo_GetSSB_Index(nr_Cell1), v_RA_PreambleIndex);
   f_NR_SS_CommonCellConfig(nr_Cell1, cads_NR_DciTriggerPDCCHOrder_REQ(nr_Cell1, -, v_PDCCH_Order));
    
   //@siclog "Step 14" siclog@
   // The UE transmits Preamble on PRACH corresponding to ra-PreambleIndex in step 13 preambleTransMax times
   t_Watchdog.start;
   alt {
        [] SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell1, -, v_RA_PreambleIndex)) {
          t_Watchdog.stop;
[bookmark: OLE_LINK7]             if (v_Repetitions==7) { //Acc. to 38.508-1, Table 4.6.3-130: RACH-ConfigGeneric, preambleTransMax = n7
                  // do nothing and exit the statement to continue with step 15
             } else {
                  v_Repetitions := v_Repetitions + 1;
                  t_Watchdog.start;
                  repeat;
             }
        }
        [] t_Watchdog.timeout {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 14");
        }
   }
    
   //@siclog "Step 15" siclog@
   //The UE transmits an RRCReestablishmentRequest message
   f_NR5GC_RRCReestablishment_Steps0To2(nr_Cell1, nr_Cell1, otherFailure);

   //@siclog "Step 16-17" siclog@
   //The SS transmits an RRCReestablishment message.
   //The UE transmita an RRCReestablishmentComplete message.
   f_NR5GC_RRCReestablishment_Steps3To7(nr_Cell1, -, -, cr_RRCReestablishmentComplete_v1610_IEs_RLF);

   //@siclog "Step 18-19" siclog@
   //The SS transmits an RRCReconfiguration message to establish SRB2 and DRB.
   //The UE transmits an RRCReconfigurationtComplete message.
   f_NR5GC_RRCReestablishment_Steps8To13(nr_Cell1, nr_Cell1);
   
   //@siclog "Step 20" siclog@
   // The SS transmits a UEInformationRequest message with rlf-ReportReq set to true.
   SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1,
                                   -,
                                   cs_NR_UEInformationRequest_RLF)); // rlf-report

   //@siclog "Step 21" siclog@
   // Check: Does the UE transmit a UEInformationResponse message with rlf-Report included?
   SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1,
                                      cr_38508_UEInformationResponse(tsc_NR_RRC_TI_Def,-,-,-,-,-,?))) -> value v_ReceivedAsp; // Table 8.1.6.1.3.6.3.3-3

   v_Nr_Rlf_Report_r16 := v_ReceivedAsp.Signalling.Rrc.Dcch.message_.messageClassExtension.c2.ueInformationResponse_r16.criticalExtensions.ueInformationResponse_r16.rlf_Report_r16;

   if (match(v_Nr_Rlf_Report_r16, cr_NR_Rlf_Report(v_RNTI_Value, -,    // c_RNTI_r16, ssbRLMConfigBitmap_r16 := tsc_SsbRLMConfigBitmap_r16
                                                   (0..127), (0..127), // rsrp, rsrq
                                                   omit, omit,          // measResultNeighCells_r16, PreviousPCellId_r16
                                                   cr_NR_RLF_Report_failedPCellId_cellGlobalId_r16(cr_CGI_Info_Logging_r16(v_PlmnID_NRCell1, v_CellIdentity_NRCell1, v_TAC_NRCell1)), // failedPCellId_r16.nrFailedPCellId_r16
                                                   omit,omit,omit,omit,-,
                                                   rlf,                  // connectionFailureType_r16
                                                   randomAccessProblem,  //rlf-Cause-r16
                                                   -,
                                                   cr_RA_InformationCommon_r16({cr_PerRAInfo_r16(cr_PerRAInfoList_r16_dlRSRPAboveThreshold)},(0..37949))))) // RA_InformationCommon_r16
    {
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 21");
    }
    else {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 21: condition not fulfilled with the received rlf-Report-r16 contents");
    }

 }//end of f_TC_8_1_6_1_3_6_NR5GC_TestBody
    



After Change
	  function f_TC_8_1_6_1_3_6_NR5GC_TestBody() runs on NR5GC_PTC
  { // @sic R5-223274 R5-225390 sic@
   var NR_SRB_COMMON_IND v_ReceivedAsp;
   var PLMN_Identity v_PlmnID_NRCell1 := f_NR_CellInfo_GetPLMN (nr_Cell1);
   var CellIdentity v_CellIdentity_NRCell1 := f_NR_CellInfo_GetCellIdentity(nr_Cell1);
   var TrackingAreaCode v_TAC_NRCell1 := f_NR_CellInfo_GetTAC (nr_Cell1);
   var RNTI_Value v_RNTI_Value := f_NR_CellInfo_GetRNTI(nr_Cell1);
   var RLF_Report_r16 v_Nr_Rlf_Report_r16;
   
   var template (value) NR_PDCCH_Order_Type v_PDCCH_Order;
   var integer v_RA_PreambleIndex := 0;
   var integer v_Repetitions := 0;
   timer t_Watchdog := 5.0;              // Local wait Timer

   //@siclog "Steps 1-11 - Void " siclog@
   
   //@siclog "Step 12" siclog@
   // The SS is configured to ignore the preambles transmitted by the UE on PRACH
   f_NR_SS_RachProcedureConfig(nr_Cell1, cs_NR_RachProcedureConfig_NoResponse);    
   
   // Enable PRACH reception indication
   f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(enable));   

   //@siclog "Step 13" siclog@
   // The SS sends a PDCCH order with ra-PreambleIndex = '000000'B.
   v_PDCCH_Order := cs_NR_PDCCH_Order(f_NR_CellInfo_GetSSB_Index(nr_Cell1), v_RA_PreambleIndex);
   f_NR_SS_CommonCellConfig(nr_Cell1, cads_NR_DciTriggerPDCCHOrder_REQ(nr_Cell1, -, v_PDCCH_Order));
    
   //@siclog "Step 14" siclog@
   // The UE transmits Preamble on PRACH corresponding to ra-PreambleIndex in step 13 preambleTransMax times
   t_Watchdog.start;
   alt {
        [] SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell1, -, ?)) {
          t_Watchdog.stop;
             if (v_Repetitions == 6) { //Acc. to 38.508-1, Table 4.6.3-130: RACH-ConfigGeneric, preambleTransMax = n7
                  // do nothing and exit the statement to continue with step 15 
             } else {
                  v_Repetitions := v_Repetitions + 1;
                  t_Watchdog.start;
                  repeat;
             }
        }
        [] t_Watchdog.timeout {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 14");
        }
   }
   f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(disable)); 
   
   //@siclog "Step 15" siclog@
   //The UE transmits an RRCReestablishmentRequest message
   f_NR5GC_RRCReestablishment_Steps0To2(nr_Cell1, nr_Cell1, otherFailure);

   //@siclog "Step 16-17" siclog@
   //The SS transmits an RRCReestablishment message.
   //The UE transmita an RRCReestablishmentComplete message.
   f_NR5GC_RRCReestablishment_Steps3To7(nr_Cell1, -, -, cr_RRCReestablishmentComplete_v1610_IEs_RLF);

   //@siclog "Step 18-19" siclog@
   //The SS transmits an RRCReconfiguration message to establish SRB2 and DRB.
   //The UE transmits an RRCReconfigurationtComplete message.
   f_NR5GC_RRCReestablishment_Steps8To13(nr_Cell1, nr_Cell1);
   
   //@siclog "Step 20" siclog@
   // The SS transmits a UEInformationRequest message with rlf-ReportReq set to true.
   SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1,
                                   -,
                                   cs_NR_UEInformationRequest_RLF)); // rlf-report

   //@siclog "Step 21" siclog@
   // Check: Does the UE transmit a UEInformationResponse message with rlf-Report included?
   SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1,
                                      cr_38508_UEInformationResponse(tsc_NR_RRC_TI_Def,-,-,-,-,-,?))) -> value v_ReceivedAsp; // Table 8.1.6.1.3.6.3.3-3

   v_Nr_Rlf_Report_r16 := v_ReceivedAsp.Signalling.Rrc.Dcch.message_.messageClassExtension.c2.ueInformationResponse_r16.criticalExtensions.ueInformationResponse_r16.rlf_Report_r16;

   if (match(v_Nr_Rlf_Report_r16, cr_NR_Rlf_Report(v_RNTI_Value, -,    // c_RNTI_r16, ssbRLMConfigBitmap_r16 := tsc_SsbRLMConfigBitmap_r16
                                                   (0..127), (0..127), // rsrp, rsrq
                                                   omit, omit,          // measResultNeighCells_r16, PreviousPCellId_r16
                                                   cr_NR_RLF_Report_failedPCellId_cellGlobalId_r16(cr_CGI_Info_Logging_r16(v_PlmnID_NRCell1, v_CellIdentity_NRCell1, v_TAC_NRCell1)), // failedPCellId_r16.nrFailedPCellId_r16
                                                   omit,omit,-,omit,-, 
                                                   rlf,                  // connectionFailureType_r16
                                                   randomAccessProblem,  //rlf-Cause-r16
                                                   -,
                                                   cr_RA_InformationCommon_r16({cr_PerRAInfo_r16(cr_PerRAInfoList_r16_dlRSRPAboveThreshold)},(0..37949))))) // RA_InformationCommon_r16
    {
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 21");
    }
    else {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 21: condition not fulfilled with the received rlf-Report-r16 contents");
    }

 }//end of f_TC_8_1_6_1_3_6_NR5GC_TestBody
    



1. [bookmark: _Toc30685521][bookmark: _Toc117597718]Change 2
	Function name
	cr_PerRAInfo_r16

	Reason for change
	The size of the reported perRAAttemptInfoList_r16 depends on the value of numberOfPreamblesSentOnSSB_r16. Current TTCN implementation will fail if the UE includes more than one entry.

	Summary of change
	Allow the UE to report additional entries

	TTCN module
	NR5GC_MDT_Specific_Templates

	MCC160 Comment
	



Before Change
	  template(present) PerRAInfo_r16 cr_PerRAInfo_r16 (template (present) PerRAAttemptInfo_r16 p_PerRAAttemptInfo_r16):=
  {
    perRASSBInfoList_r16 := {
      ssb_Index_r16                     := ?,
[bookmark: OLE_LINK3]      numberOfPreamblesSentOnSSB_r16    := ?,
      perRAAttemptInfoList_r16          := {p_PerRAAttemptInfo_r16}
    }
  };



After Change
	  template(present) PerRAInfo_r16 cr_PerRAInfo_r16 (template (present) PerRAAttemptInfo_r16 p_PerRAAttemptInfo_r16):=
  {
    perRASSBInfoList_r16 := {
      ssb_Index_r16                     := ?,
      numberOfPreamblesSentOnSSB_r16    := ?,
[bookmark: OLE_LINK2]      perRAAttemptInfoList_r16          := {p_PerRAAttemptInfo_r16,*} 
    }
  };



1. [bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665][bookmark: _Toc117597719]Branches executed
[bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666]The test case was executed on NR band n41 using NR ciphering nea1 and integrity algorithm nia1 
3 [bookmark: _Toc117597720]Execution Log Files
3.1 [bookmark: _Toc117597721]MediaTek MT6895
The MediaTek MT6895 UE passed this test case on Keysight S8704A Protocol Conformance Toolset. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
TC_8_1_6_1_3_6_LOG.html
· PICS/PIXIT parameter file:
       TC_8_1_6_1_3_6_PIXIT.xml


4 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc117597722]References 
	[1]
	R5s221240 : NR5GC FR1 : Supporting information for addition of NR5GC Rel-16 SON-MDT test case 8.1.6.1.3.6



